Bit Mirror Routine; Series 32000® Graphics Note 7

National Semiconductor
Application Note 530
Dave Rand
May 1988

1.0 INTRODUCTION

The bit mirror routine is designed to reorder the bits in an image. The bits are swapped around a fixed point, that being one half of the size of the data, as is shown for the byte mirror below. These routines can be used for conversion of 68000 based data.

2.0 DESCRIPTION

	Bit Number								Hex Value
	7	6	5	4	3	2	1	0	
Source	1	0	1	1	0	0	1	0	B2
Result of Mirror	0	1	0	0	1	1	0	1	4D

The ''mirror'', in this case, is between bits 3 and 4.

Several different algorithms are available for the mirror operation. The best algorithm to mirror a byte takes 20 clocks on a NS32016 (about 2.5 clocks per bit), and uses a 256 byte table to do the mirror operation. The table is reproduced at the end of this document. To perform a byte mirror, the following code may be used. The byte to be mirrored is in R0, and the destination is to be R1.

```
MOVB mirtab[r0:b],r1     #Mirror a byte
```

TL/EE/9700–1

An extension of this algorithm is used to mirror larger amounts of data. To mirror a 32-bit block of data from one location to another, the following code may be used. Register R0 points to the source block, register R1 points to the destination. R2 is used as a temporary value.

```
MOVZBD    0(r0),r2              #get first byte
MOVB      mirtab[r2:b],3(r1)    #store in last place
MOVB      1(r0),r2              #get next byte
MOVB      mirtab[r2:b],2(r1)    #store in next place
MOVB      2(r0),r2              #get the third byte
MOVB      mirtab[r2:b],1(r1)    #store in next place
MOVB      3(r0),r2              #get the last byte
MOVB      mirtab[r2:b],0(r1)    #first place
```

TL/EE/9700–2

This code uses 33 bytes of memory, and just 169 clocks to execute. Larger blocks of data can be mirrored with this method as well, with each additional byte taking about 40 clocks.

Registers can also be mirrored with this method, with just a few more instructions. To mirror R0 to R1, for example, the following code could be used. R2 is used as a temporary variable.

```
MOVZBD    r0,r2           #get lsbyte
MOVB      mirtab[r2:b],r1 #mirror the byte
LSHD      $8,r1           #move into higher byte of destination
LSHD      $-8,r0          #and of source
MOVB      r0,r2           #get lsbyte
MOVB      mirtab[r2:b],r1 #mirror the byte
LSHD      $8,r1           #move into higher byte of destination
LSHD      $-8,r0          #and of source
MOVB      r0,r2           #get lsbyte
MOVB      mirtab[r2:b],r1 #mirror the byte
LSHD      $8,r1           #move into higher byte of destination
LSHD      $-8,r0          #and of source
MOVB      r0,r2           #get lsbyte
MOVB      mirtab[r2:b],r1 #mirror the byte
```

TL/EE/9700–3

RRD-B30M105/Printed in U. S. A.

This code occupies 49 bytes, and executes in 286 clocks on an NS32016.

If space is at a premium, a shorter table may be used, at the expense of time. Each nibble (4 bits) instead of each byte is processed. This means that the table only requires 16 entries. To mirror a byte in R0 to R1, the following code can be used. R2 is used as a temporary variable.

```
MOVB    r0,r2               #get lsbyte
ANDD    $15,r2              #mask to get ls nibble
MOVB    mirtb16[r2:b],r1    #mirror the nibble
LSHD    $4,r1               #high nibble of destination
LSHD    $-4,r0              #and of source
MOVB    r0,r2               #get lsbyte
ANDD    $15,r2              #mask to get ls nibble
ORB     mirtb16[r2:b],r1    #mirror the nibble
```

TL/EE/9700–4

This code requires 32 bytes of memory, and executes in 125 clock cycles on an NS32016. A slightly faster time (100 clocks) may be obtained by adding a second table for the high nibble, and eliminating the LSHD 4,r1 instruction.

TABLES

MIRTAB is a table of all possible mirror values of 8 bits, or 256 bytes. MIRTB16 is a table of all possible mirror values of 4 bits, or 16 bytes. These tables should be aligned for best performance. They may reside in code (PC relative), or data (SB relative) space.

```
mirtab:
        .byte   0x00,0x80,0x40,0xc0,0x20,0xa0,0x60,0xe0,0x10,0x90,0x50
        .byte   0xd0,0x30,0xb0,0x70,0xf0
        .byte   0x08,0x88,0x48,0xc8,0x28,0xa8,0x68,0xe8,0x18,0x98,0x58
        .byte   0xd8,0x38,0xb8,0x78,0xf8
        .byte   0x04,0x84,0x44,0xc4,0x24,0xa4,0x64,0xe4,0x14,0x94,0x54
        .byte   0xd4,0x34,0xb4,0x74,0xf4
        .byte   0x0c,0x8c,0x4c,0xcc,0x2c,0xac,0x6c,0xec,0x1c,0x9c,0x5c
        .byte   0xdc,0x3c,0xbc,0x7c,0xfc
        .byte   0x02,0x82,0x42,0xc2,0x22,0xa2,0x62,0xe2,0x12,0x92,0x52
        .byte   0xd2,0x32,0xb2,0x72,0xf2
        .byte   0x0a,0x8a,0x4a,0xca,0x2a,0xaa,0x6a,0xea,0x1a,0x9a,0x5a
        .byte   0xda,0x3a,0xba,0x7a,0xfa
        .byte   0x06,0x86,0x46,0xc6,0x26,0xa6,0x66,0xe6,0x16,0x96,0x56
        .byte   0xd6,0x36,0xb6,0x76,0xf6
        .byte   0x0e,0x8e,0x4e,0xce,0x2e,0xae,0x6e,0xee,0x1e,0x9e,0x5e
        .byte   0xde,0x3e,0xbe,0x7e,0xfe
        .byte   0x01,0x81,0x41,0xc1,0x21,0xa1,0x61,0xe1,0x11,0x91,0x51
        .byte   0xd1,0x31,0xb1,0x71,0xf1
        .byte   0x09,0x89,0x49,0xc9,0x29,0xa9,0x69,0xe9,0x19,0x99,0x59
        .byte   0xd9,0x39,0xb9,0x79,0xf9
        .byte   0x05,0x85,0x45,0xc5,0x25,0xa5,0x65,0xe5,0x15,0x95,0x55
        .byte   0xd5,0x35,0xb5,0x75,0xf5
        .byte   0x0d,0x8d,0x4d,0xcd,0x2d,0xad,0x6d,0xed,0x1d,0x9d,0x5d
        .byte   0xdd,0x3d,0xbd,0x7d,0xfd
        .byte   0x03,0x83,0x43,0xc3,0x23,0xa3,0x63,0xe3,0x13,0x93,0x53
        .byte   0xd3,0x33,0xb3,0x73,0xf3
        .byte   0x0b,0x8b,0x4b,0xcb,0x2b,0xab,0x6b,0xeb,0x1b,0x9b,0x5b
        .byte   0xdb,0x3b,0xbb,0x7b,0xfb
        .byte   0x07,0x87,0x47,0xc7,0x27,0xa7,0x67,0xe7,0x17,0x97,0x57
        .byte   0xd7,0x37,0xb7,0x77,0xf7
        .byte   0x0f,0x8f,0x4f,0xcf,0x2f,0xaf,0x6f,0xef,0x1f,0x9f,0x5f
        .byte   0xdf,0x3f,0xbf,0x7f,0xff

mirtb16:
        .byte   0x0,0x8,0x4,0xc,0x2,0xa,0x6,0xe,0x1,0x9,0x5
        .byte   0xd,0x3,0xb,0x7,0xf
```

TL/EE/9700–5



Lit. # 100529

National Semiconductor Corporation
1111 West Bardin Road
Arlington, TX 76017
Tel: 1(800) 272-9959
Fax: 1(800) 737-7018

National Semiconductor Europe
Fax: (+49) 0-180-530 85 86
Email: cnjwge@tevm2.nsc.com
Deutsch Tel: (+49) 0-180-530 85 85
English Tel: (+49) 0-180-532 78 32
Français Tel: (+49) 0-180-532 93 58
Italiano Tel: (+49) 0-180-534 16 80

National Semiconductor Hong Kong Ltd.
13th Floor, Straight Block,
Ocean Centre, 5 Canton Rd.
Tsimshatsui, Kowloon
Hong Kong
Tel: (852) 2737-1600
Fax: (852) 2736-9960

National Semiconductor Japan Ltd.
Tel: 81-043-299-2309
Fax: 81-043-299-2408